Swamp Cooler

Introduction

A swamp cooler reduces the temperature of air through evaporative cooling.

Hot dry air at 40°C and 10% relative humidity passes through a swamp cooler. Water is added as the air passes through a series of wicks and the mixture exits at 27 °C. This application calculates

- the relative humidity of the air exiting the cooler,
- the mass of water added,
- and the lowest achievable temperature at the outlet,

and plots the process on a psychrometric chart.

Calculations

```
> restart :
  with( ThermophysicalData ) : with( plots ) :
> T1 := 40 + 273.15 :
  R1 := 0.1 :
  T2 := 27 + 273.15 :
> W1 := Property( W, HumidAir, Tdb = T1, pressure = 101325, R = 0.1 )
```

$$W1 := 0.00458815922768428600$$

The lowest possible temperature and humidty ratio is the wet-bulb temperature at the inlet conditions.

```
> T_lowest := Property( Twb, HumidAir, Tdb = T1, pressure = 101325, R = 0.1 );
```

$$T\_lowest := 291.697570144506244$$

```
> W_lowest := Property( W, HumidAir, Tdb = T_lowest, pressure = 101325, R = 1 );
```

$$W\_lowest := 0.0134556886875402664$$

The outlet relative humidity and humidity ratio are

```
> R2 := Property( R, HumidAir, Tdb = T2, pressure = 101325, Twb = T_lowest );
```

$$R2 := 0.444247577124856352$$

```
> W2 := Property( W, HumidAir, Tdb = T2, pressure = 101325, Twb = T_lowest );
```

$$W2 := 0.00992726950779089859$$

Hence the mass of water added per mass of dry air is

```
> W2 − W1
```

$$0.005339110280$$

```
> mixingPlotPoints := plots:-pointplot( [ [T1, W1], [T2, W2], [T_lowest, W_lowest] ], connect = false,
```

```
    symbol = solidcircle, symbolsize = 15, color = RGB(150/225, 40/255, 27/255)) :
  mixingPlotLines := plots:-pointplot([[T1, W1], [T2, W2], [T_lowest, W_lowest], [T_lowest, 0]],
    connect = true, thickness = 2) :
```

```
> display(PsychrometricChart( ), mixingPlotPoints, mixingPlotLines)
```